

WANG

MATHEMATICS
GENERAL PROGRAM LIBRARY
LISTINGS
GLBR 22

Jack Jarvis & Company, Inc.
707 S. W. Washington Street
Portland, Oregon 97205
Telephone: (503) 224-7838

SYSTEM 2200





.

.



.

.



Jack Jarvis & Company, Inc.
707 S. W. Washington Street
Portland, Oregon 97205
Telephone: (503) 221-7777

2200 General Library Listings Mathematics GLBR 22

© Wang Laboratories, Inc., 1973



LABORATORIES, INC.

836 NORTH STREET, TEWKSBURY, MASSACHUSETTS 01876, TEL. (617) 851-4111, TWX 710 343-6769, TELEX 94-7421

Warranty Disclaimer and Consequential Damages

The programming staff of Wang Laboratories, Inc. has taken due care to prepare this program package, including research, development, and testing to ascertain its effectiveness. Wang Laboratories, Inc. and its subsidiaries make no expressed or implied warranty of any kind with regard to this program package and its related material. In no event shall Wang Laboratories, Inc. or its subsidiaries be liable for incidental or consequential damages in connection with or arising out of the furnishing, performance or use of any of its programs.

Proprietary Rights

This program package is a company proprietary item. Tape cassettes of programs may not be reproduced without the written consent of Wang Laboratories, Inc.

Payment Address

The fee for a package is to be remitted directly to Wang Laboratories, Inc. at 836 North Street, Tewksbury, Massachusetts 01876. No agent or representative is charged with the authority to accept payment on behalf of the administrative offices of Wang, Laboratories, Inc.

TABLE OF CONTENTS

Tape I

BLOCK	PROGRAM	PAGE
1	ROOTS OF A QUADRATIC	1
2	ROOTS OF A POLYNOMIAL.....	2
3	HALF-INTERVAL SEARCH FOR ROOTS.....	3
4	REAL ROOTS OF A POLYNOMIAL.....	4
5	SIMPSON'S RULE	5
6	NUMERICAL INTEGRATION (ROMBERG'S METHOD).....	6
7	RUNGE-KUTTA	7
8	GAUSSIAN QUADRATURE (20-point).....	8
9	DERIVATIVE (DIFFERENCE QUOTIENTS).....	9
10	MATRIX INVERSION (GAUSS-JORDAN ELIMINATION METHOD).....	10
11	MATRIX INVERSION (GAUSS-JORDAN DONE IN PLACE).....	11
12	EIGENVALUE AND EIGENVECTOR	12
13	VECTOR OPERATIONS	13
14	VECTOR ANALYSIS	14
15	SOLUTION OF SIMULTANEOUS EQUATIONS (GAUSS-JORDAN).....	15
16	MATRIX ADDITION, SUBTRACTION AND SCALAR MULTIPLICATION.....	16
17	MATRIX MULTIPLICATION	17
18	SOLUTION OF SIMULTANEOUS EQUATIONS	18
19	LINEAR PROGRAMMING	19
20	COMPLEX DETERMINANT.....	20
21	HYPERBOLIC FUNCTIONS & INVERSE HYPERBOLICS.....	21
22	SIN, COS, TAN, SINH, COSH, TANH - COMPLEX ARGUMENTS.....	22
23	ANGLE CONVERSION I	23
24	ANGLE CONVERSION II.....	24
25	TRIGONOMETRIC POLYNOMIAL.....	25
26	PLANE TRIANGLE SOLUTION.....	26
27	COORDINATE CHANGE	28
28	AREA OF RECTILINEAR SURFACE.....	29
29	LINEAR INTERPOLATION.....	30
30	LAGRANGIAN INTERPOLATION.....	31
31	GREATEST COMMON DIVISOR.....	32
32	PRIME FACTORIZATION OF AN INTEGER.....	33
33	PERMUTATIONS AND COMBINATIONS.....	34
34	LOG B TO BASE A	35
35	SECOND DEGREE EQUATION I	36
36	EXPLICIT SECOND DEGREE EQUATION	37
37	SECOND DEGREE EQUATION II	38
38	ALGEBRA OF COMPLEX NUMBERS	39
39	HYPERGEOMETRIC FUNCTION	40
40	SQUARE ROOT OF A COMPLEX NUMBER	41
41	BESSEL FUNCTION	42
42	GAMMA FUNCTION	43
43	FOURIER ANALYSIS (DEFINED FUNCTION)	44
44	FOURIER ANALYSIS (TABULATED FUNCTION)	45



.

.



.

.



```
10 PRINT HEX(03)
20 PRINT "ROOTS OF QUADRATIC"
25 PRINT "A1, A2, A3"
30 INPUT A1, A2, A3
40 X1=-A2/(2*A1)
50 D=X1*X1-A3/A1
60 IF D<0THEN 90
70 X2=X1-SQR(D):X1=X1+SQR(D)
80 I1=0:I2=0:GOTO 100
90 X2=X1:I1=SQR(-D):I2=-SQR(-D)
100 PRINT "X1 REAL =":X1,"X2 REAL =":X2
110 IF I1=0 THEN 140
120 PRINT "X1 IMAG =":I1,"X2 IMAG =":I2
130 PRINT "MORE INPUT (1=YES,0=NO)"
132 INPUT A:IF A=1THEN 20
135 END
140 IF I2=0 THEN 130
150 GOTO 120
160 END
```

```

1 DIM B(22),E(22),W(22),C(22)
5 E0=1E-12: E1=1E-4: E4=1E-20: K1=100: S5=0
10 PRINT "DEGREE OF POLYNOMIAL?":INPUT N
15 PRINT "ENTER COEFF. : A0,A1,...,AN 1/LINE"
20 FOR I=1 TO N+1:INPUT B(N-I+2)
25 NEXT I:IF ABS(B(1))>=E0 THEN 35
30 PRINT "A(N) ZERO OR NEARLY ZERO. CALCULATIONS END. ":GOTO 999

35 PRINT :PRINT "ROOTS:":IF N<=2 THEN 350
40 B(N+2)=0: N1=2*INT((N+1)/2)
45 FOR M1=1 TO N1/2: P,Q=1
50 FOR K=1 TO K1:FOR L=1 TO K1
55 FOR I=1 TO N1+1: C(I)=B(I):NEXT I
60 FOR J=N1-2 TO N1-4 STEP -2:FOR I=1 TO J+1
65 C(I+1)=C(I+1)-P*C(I): C(I+2)=C(I+2)-Q*C(I)
70 NEXT I:NEXT J
75 R0=C(N1+1): R1=C(N1)
80 S0=C(N1-1): S1=C(N1-2)
85 V0=-Q*S1: V1=S0-S1*P
90 D0=V1+S0-V0*S1:IF ABS(D0)>=E4 THEN 110
95 P=P+5: Q=Q+5
100 NEXT L
110 D1=S0*R1-S1*R0: D2=R0*V1-V0*R1
120 P1=D1/D0: Q1=D2/D0: P=P+P1: Q=Q+Q1
130 IF ABS(R0)>=E1 THEN 150:IF ABS(R1)>=E1 THEN 150
140 E(M1)=1:GOTO 210
150 IF ABS(P1)>=E1 THEN 170:IF ABS(Q1)>=E1 THEN 170
160 E(M1)=2:GOTO 210
170 IF P=0 THEN 180:IF ABS(P1/P)>=E1 THEN 200
180 IF Q=0 THEN 200:IF ABS(Q1/Q)>=E1 THEN 200
190 E(M1)=3:GOTO 210
200 NEXT K: E(M1)=4
210 S=-P/2: T=S*S-Q
220 IF T<0 THEN 240: T=SQR(T): W(M1)=1
230 PRINT :PRINT S+T:PRINT S-T:GOTO 260
240 W(M1)=-1: T=SQR(-T)
250 PRINT :PRINT S;"+I*":T:PRINT S:"-I*":T
260 IF E(M1)=4 THEN 999
270 FOR J=1 TO N1-1
280 B(J+1)=B(J+1)-P*B(J): B(J+2)=B(J+2)-Q*B(J)
290 NEXT J
300 N1=N1-2:IF N1>1 THEN 310:GOTO 999
310 IF N1=3 THEN 340
320 M1=M1+1: E(M1)=1
330 P=B(2)/B(1): Q=B(3)/B(1):GOTO 210
340 NEXT M1
350 IF N=2 THEN 370
360 PRINT -B(2)/B(1):GOTO 999
370 B(3)=B(2)*B(2)-4*B(1)*B(3)
380 S=-B(2)/2/B(1): T=SQR(ABS(B(3)))/2/B(1)
390 M1,E(4)=4:IF SGN(B(3))<0 THEN 250:GOTO 230
999 END

```


ROOTS

BLOCK NO.: 3

```
100 DIM D(3)
105 PRINT "INTERVAL (LOWER LIMIT, UPPER LIMIT)?"
110 INPUT A,B
112 S=SGN(FNC(A))
115 T=SGN(FNC(B))
120 PRINT
130 IF S*T=0 THEN 800
140 IF S*T<0 THEN 500
200 FOR I=1 TO 1000
210     X=A+RND(Z)*(B-A)
220     V=SGN(FNC(X))
230     IF V=0 THEN 900
240     IF S*V<0 THEN 400
250 NEXT I
260 PRINT "NO CHANGE OF SIGN FOUND"
265 PRINT
270 GOTO 100
400 B=X
500 D(2+S)=A
510 D(2-S)=B
600 X=(D(1)+D(3))/2
610 U=SGN(FNC(X))
620 IF U=0 THEN 900
630 D(2+U)=X
640 IF ABS(D(1)-D(3))/(ABS(D(1))+ABS(D(3)))<5E-6 THEN 900
650 GOTO 600
800 IF S=0 THEN 850
810 X=T
820 GOTO 900
850 X=S
900 PRINT "ONE ROOT AT ",X
905 PRINT
910 GOTO 100
999 END
```

```
2 PRINT HEX(03)
5 DIM A(99)
10 PRINT "INPUT DEGREE OF POLYNOMIAL, N"
20 INPUT N
22 PRINT "INPUT COEFFICIENTS A(1),...,A(N+1) 4/LINE"
25 FOR I=0 TO INT((N+4)/4)-1
30 INPUT A(4*I+1),A(4*I+2),A(4*I+3),A(4*I+4)
35 NEXT I
40 PRINT "INPUT YOUR ESTIMATE OF ROOT"
45 INPUT X
50 FOR J=1 TO 100
55 F=A(N+1)
60 FOR I=1 TO N
70 T=T+X^2
80 F=F+A(I)*X^(N-I+1)
90 NEXT I
100 G=A(N)
110 FOR I=1 TO N-1
120 G=G+(N-I+1)*A(I)*X^(N-I)
130 NEXT I
140 I=X
150 X=X-F/G
160 IF ABS(X)+ABS(I)=0 THEN 900
165 IF X=0 THEN 200
170 IF ABS(1-I/X)<10E-8 THEN 900
200 NEXT J
210 PRINT "AFTER 100 ITERATIONS, NO CONVERGENCE"
220 GOTO 999
900 PRINT "ROOT IS: ",X
999 PRINT "ANOTHER ROOT (1=YES,0=NO)":INPUT L:IF L=1THEN 40:END
```

```
15 READ Z, A, B, D
20 S, T=0
30 IF Z=1 THEN 60
40 READ Y1
50 GOTO 80
60 Y1=FNC(A)
70 Y2=FNC(B)
80 FOR I=1 TO (B-A)/D+.5
90 IF Z=1 THEN 120
100 READ Y
110 GOTO 130
120 Y=FNC(A+I*D)
130 IF I/2=INT(I/2) THEN 160
140 S=S+Y
150 GOTO 170
160 T=T+Y
170 NEXT I
180 IF Z=1 THEN 200
190 READ Y2
200 PRINT "INTEGRAL="; D/3*(Y1+4*S+2*T+Y2)
999 END
```

(ROMBERG'S METHOD)

BLOCK NO.: 6

```
5 DIM T(8)
10 PRINT "INPUT 'LOWER LIMIT, UPPER LIMIT' (TO END PROGRAM INPUT
"
11 PRINT "EQUAL LIMITS)"
20 INPUT X1,X2
30 IF X2=X1 THEN 999
40 L=X2-X1
50 Y1=FNC(X1)
60 Y=FNC(X2)
70 T(1)=(Y1+Y)/2
80 N=1
90 FOR H=1 TO 7
100 U=0
110 M=L/(2*N)
120 FOR J=1 TO 2*N-1 STEP 2
130 Y=FNC(X1+J*M)
140 U=U+Y
150 NEXT J
160 T(H+1)=(U/N+T(H))/2
170 F=1
180 FOR J=H TO 1 STEP -1
190 F=F*4
200 T(J)=T(J+1)+(T(J+1)-T(J))/(F-1)
210 NEXT J
220 N=2*N
230 I2=T(1)*L
240 IF H=1 THEN 260
250 IF ABS(I1-I2)<=1E-4*ABS(I2) THEN 400
260 I1=I2
270 NEXT H
280 PRINT "UNABLE TO COMPUTE INTEGRAL TO 4 SIGNIFICANT DIGITS. "
290 PRINT "CLOSEST APPROXIMATION IS: "; I2
295 PRINT
300 PRINT "INPUT 'LOWER LIMIT, UPPER LIMIT'"
310 GOTO 20
400 PRINT "INTEGRAL= "; I2
410 GOTO 295
999 END
```

```
1 DIM X(20),F(10),K(4,10)
3 GOTO 100
5 REM --SUBROUTINE
99 RETURN
210 PRINT "T",
220 READ N,T
230 FOR I=1 TO N
240 READ X(I)
250 PRINT "X": I,
260 NEXT I
270 READ D,B
280 A=T
285 FOR J=A TO B STEP D
286 PRINT
288 PRINT T,
290 FOR I=1 TO N
292 PRINT X(I),
294 NEXT I
300 FOR I=1 TO N
310 X(N+I)=X(I)
320 NEXT I
330 GOSUB 5
340 FOR I=1 TO N
350 K(1, I)=F(I)
360 NEXT I
370 D1=D/2
380 T=T+D1
390 FOR L=1 TO 3
400 T=T+INT(L/3)*D1
410 FOR I=1 TO N
420 X(I)=X(N+I)+D1*K(L, I)*INT(L/3+1)
430 NEXT I
440 GOSUB 5
450 FOR I=1 TO N
460 K(L+1, I)=F(I)
470 NEXT I
480 NEXT L
610 FOR I=1 TO N
620 X(I)=X(N+I)+D/6*(K(1, I)+2*K(2, I)+2*K(3, I)+K(4, I))
630 NEXT I
640 NEXT J
999 END
```

(20-POINT)

BLOCK NO.: 8

```
10 PRINT "ENTER LOWER LIMIT OF INTEGRATION, UPPER LIMIT OF INTEGRATION"
20 INPUT A, B
30 PRINT "ENTER NO. OF SUBINTERVALS"
40 INPUT K
50 C=(B-A)/K/2
110 D=A+C
115 T=0
120 FOR J=1 TO K
130 S=0
140 FOR I=1 TO 10
150 READ X, W
160 S=S+W*(FNC(C*X+D)+FNC(D-C*X))
170 NEXT I
175 RESTORE
180 T=T+S*C
185 D=D+2*C
190 NEXT J
200 PRINT "INTEGRAL= "; T
210 PRINT
220 PRINT "CHANGE NUMBER OF SUBINTERVALS? (1/--YES, 0/--NO)"
230 INPUT I
240 IF I=1 THEN 30
250 PRINT "NEW INTEGRATION LIMITS? (1/--YES, 0/--NO)"
260 INPUT I
270 IF I=1 THEN 10
300 DATA 76526521E-9, .15275339, .22778585, .14917299, .37370609, .14
209611
310 DATA .510867, .13168864, .63605368, .11819453, .74633191, .101930
12
320 DATA .83911697, 83276742E-9, .91223443, 62672048E-9, .96397193
330 DATA 4060143E-8, .9931286, 17614007E-9
999 END
```

(DIFFERENCE QUOTIENTS)

BLOCK NO.: 9

```
5 PRINT "ENTER VALUE OF X. (TO END PROGRAM INPUT 99999)"
10 INPUT A
12 IF A=99999 THEN 90
15 PRINT
20 PRINT " X", "DIFFERENCE QUOTIENT"
25 D=0
30 FOR N=1 TO 10
35 D1=D
40 X=A+.5*N
45 D=(FNC(X)-FNC(A))/(X-A)
50 PRINT INT(X*1E4+.5)/1E4, D
55 NEXT N
60 PRINT
65 PRINT "DERIVATIVE AT X="; A; " IS: "; 2*D-D1
70 PRINT
80 PRINT "ENTER VALUE OF X"
85 GOTO 10
90 END
```

JORDAN ELIMINATION METHOD)

```
5 DIM A(6,7),Z(6,6)
10 PRINT "INPUT N":INPUT N
20 PRINT "INPUT MATRIX"
30 FOR S=1 TO N
40 INPUT A(S,1),A(S,2),A(S,3),A(S,4),A(S,5),A(S,6),A(S,7)
60 Z(S,S)=1:NEXT S
70 FOR S=1 TO N
80 FOR T=S TO N:IF A(T,S)<>0 THEN 100:NEXT T
90 PRINT "MATRIX SINGULAR":GOTO 999
100 GOSUB 500
110 C=1/A(S,S):GOSUB 600
120 FOR T=1 TO N:IF T=S THEN 140
130 C=-A(T,S):GOSUB 700
140 NEXT T:NEXT S:GOTO 800
500 FOR J=1 TO N
510 B=A(S,J): A(S,J)=A(T,J): A(T,J)=B
520 B=Z(S,J): Z(S,J)=Z(T,J): Z(T,J)=B
530 NEXT J:RETURN
600 FOR J=1 TO N
610 A(S,J)=C*A(S,J): Z(S,J)=C*Z(S,J)
620 NEXT J:RETURN
700 FOR J=1 TO N
710 A(T,J)=A(T,J)+C*A(S,J): Z(T,J)=Z(T,J)+C*Z(S,J)
720 NEXT J:RETURN
800 PRINT
810 FOR I=1 TO N:FOR J=1 TO N
820 PRINT Z(I,J):NEXT J:PRINT :NEXT I
999 END
```


JORDAN DONE IN PLACE)

BLOCK NO. 11

```
5 DIM A(10,11),V(10)
10 PRINT "ENTER N":INPUT N
15 PRINT "ENTER MATRIX"
20 FOR S=1 TO N
25 INPUT A(S,1),A(S,2),A(S,3),A(S,4),A(S,5),A(S,6),A(S,7),A(S,8)
    A(S,9),A(S,10),A(S,11)
30 NEXT S
35 FOR S=1 TO N
40 FOR T=S TO N:IF A(T,S)=0 THEN 50:NEXT T
45 PRINT "MATRIX SINGULAR":GOTO 999
50 GOSUB 500
55 A(S,S)=1/A(S,S):GOSUB 600
60 FOR T=1 TO N:IF T=S THEN 70
65 B=-A(T,S): A(T,S)=0:GOSUB 700
70 NEXT T:NEXT S
100 FOR S=N TO 1 STEP -1:IF V(S)=S THEN 140
110 FOR J=1 TO N
120 B=A(J,S): A(J,S)=A(J,V(S)): A(J,V(S))=B
130 NEXT J
140 NEXT S:GOTO 800
500 FOR J=1 TO N
510 B=A(S,J): A(S,J)=A(T,J): A(T,J)=B
520 NEXT J: V(S)=T:RETURN
600 FOR J=1 TO N:IF J=S THEN 620
610 A(S,J)=A(S,S)*A(S,J)
620 NEXT J:RETURN
700 FOR J=1 TO N
710 A(T,J)=A(T,J)+B*A(S,J)
720 NEXT J:RETURN
800 FOR I=1 TO N:PRINT :FOR J=1 TO N:PRINT A(I,J),
810 NEXT J:PRINT :NEXT I
999 END
```

```
10 DIM A(5,6),S(5,5)
20 R=.00001
30 PRINT "INPUT N":INPUT N
40 PRINT "INPUT MATRIX"
50 FOR I=1 TO N
60 INPUT A(I,1),A(I,2),A(I,3),A(I,4),A(I,5),A(I,6),A(I,7),A(I,8)
,      A(I,9),A(I,10)
75 NEXT I
80 FOR I=1 TO N: S(I,I)=1:NEXT I
90 FOR I=2 TO N:FOR J=1 TO I-1
100 I1=I1+2*A(I,J)^2
110 NEXT J:NEXT I
130 N1=SQR(I1): N2=(R/N)*N1: T=N1
140 T=T/N
150 FOR Q=2 TO N:FOR P=1 TO Q-1
160 IF ABS(A(P,Q))<=T THEN 320
170 I2=1
180 V1=A(P,P): V2=A(P,Q): V3=A(Q,Q)
190 M1=(V1-V3)*.5:IF M1<>0 THEN 210
200 W=-1:GOTO 220
210 W=-SGN(M1)*V2/SQR(V2^2+M1^2)
220 T1=W/SQR(2*(1+SQR(1-W/2))): T2=T1^2
230 C1=SQR(1-T2): C2=C1^2: T3=T1*C1
240 FOR I=1 TO N
250 I1=A(I,P)*C1-A(I,Q)*T1: A(I,Q)=A(I,P)*T1+A(I,Q)*C1
260 A(I,P)=I1: I1=S(I,P)*C1-S(I,Q)*T1
270 S(I,Q)=S(I,P)*T1+S(I,Q)*C1:S(I,P)=I1
280 NEXT I
290 FOR I=1 TO N: A(P,I)=A(I,P): A(Q,I)=A(I,Q):NEXT I
300 A(P,P)=V1*C2+V3*T2-2*V2*T3: A(Q,Q)=V1*T2+V3*C2+2*V2*T3
310 A(P,Q)=(V1-V3)*T3+V2*(C2-T2): A(Q,P)=A(P,Q)
320 NEXT P
330 NEXT Q
340 IF I2<>1 THEN 360
350 I2=0:GOTO 150
360 IF T>N2 THEN 140
370 PRINT "EIGENVALUE", "EIGENVECTOR":PRINT
380 FOR I=1 TO N:PRINT A(I,I),S(I,I)
390 FOR J=2 TO N:PRINT " ",S(J,I):NEXT J
400 PRINT :PRINT :NEXT I
999 END
```

```
5 PRINT "VECTOR A?"
10 INPUT X1, Y1, Z1
15 PRINT "VECTOR B?"
20 INPUT X2, Y2, Z2
25 PRINT
30 PRINT "A+B=( "; X1+X2; ", "; Y1+Y2; ", "; Z1+Z2; ")"
35 PRINT "A-B=( "; X1-X2; ", "; Y1-Y2; ", "; Z1-Z2; ")"
40 PRINT "AB="; X1*X2+Y1*Y2+Z1*Z2
45 PRINT "AXB=( "; Y1*Z2-Z1*Y2; ", "; Z1*X2-X1*Z2; ", "; X1*Y2-Y1*X2; ")"

50 PRINT
55 PRINT "MORE INPUT? (1=YES, 0=NO)"
60 INPUT X1
65 PRINT
70 IF X1=1 THEN 5
75 END
```

```
1 SELECT R
5 DIM X(6), M(4), A(3)
10 PRINT
15 PRINT "INPUT VECTORS 1/LINE (X, Y, Z, CARRIAGE RETURN)"
30 INPUT X(1), X(2), X(3)
40 INPUT X(4), X(5), X(6)
45 PRINT
50 FOR I=1 TO 4 STEP 3
60 M(I)=SQR(X(I)^2+X(I+1)^2+X(I+2)^2)
70 IF M(I)=0 THEN 180
80 PRINT "FOR VECTOR "; INT(I/4)+1
90 PRINT "  MAGNITUDE="; M(I)
100 FOR J=1 TO 3
110 B=X(I+J-1)/M(I)
120 A(J)=ATN(SQR(1-B^2)/B)*57.29578
130 NEXT J
140 PRINT "  ANGLE BETW. VECTOR AND X-AXIS="; A(1)
150 PRINT "  ANGLE BETW. VECTOR AND Y-AXIS="; A(2)
160 PRINT "  ANGLE BETW. VECTOR AND Z-AXIS="; A(3)
170 PRINT
180 NEXT I
185 B=0
190 IF M(1)=0 THEN 260
200 IF M(4)=0 THEN 260
210 B=(X(1)*X(4)+X(2)*X(5)+X(3)*X(6))/M(1)/M(4)
220 IF B<>0 THEN 250
230 B=90
240 GOTO 260
250 B=ATN(SQR(1-B^2)/B)*57.29578
260 PRINT "ANGLE BETW. VECTORS="; B
280 PRINT
290 PRINT "MORE INPUT? (1=YES, 0=NO)"
300 INPUT B
310 IF B=1 THEN 10
320 SELECT D :END
```

EQUATIONS (GAUSS-JORDAN)

```
5 DIM A(7,9)
10 PRINT "NO. OF UNKNOWN?"
20 INPUT N
30 PRINT "INPUT AUGMENTED COEFFICIENT MATRIX"
100 FOR I=1 TO N
120 INPUT A(I,1),A(I,2),A(I,3),A(I,4),A(I,5),A(I,6),A(I,7),
      A(I,8),A(I,9)
140 NEXT I
150 PRINT
210 FOR S=1 TO N
220 FOR T=S TO N
230 IF A(T,S) <> 0 THEN 240
235 NEXT T
237 PRINT "NO UNIQUE SOLUTION"
238 GOTO 999
240 GOSUB 510
250 C=1/A(S,S)
260 GOSUB 610
270 FOR T=1 TO N
275 IF T=S THEN 300
280 C=-A(T,S)
290 GOSUB 710
300 NEXT T
305 NEXT S
310 GOTO 800
510 FOR J=1 TO N+1
520 B=A(S,J)
530 A(S,J)=A(T,J)
540 A(T,J)=B
550 NEXT J
560 RETURN
610 FOR J=1 TO N+1
620 A(S,J)=C*A(S,J)
630 NEXT J
640 RETURN
710 FOR J=1 TO N+1
720 A(T,J)=A(T,J)+C*A(S,J)
730 NEXT J
740 RETURN
800 FOR T=1 TO N
810 PRINT "X(" T; ")=" A(T,N+1)
820 NEXT T
999 END
```

AND SCALAR MULTIPLICATION

```
5 DIM A(10,10),X(10)
10 PRINT "INPUT 1 (SCALAR MULT.) OR 2 (ADD) OR 3 (SUBTRACT)"
15 INPUT Z:IF Z>1 THEN 20:PRINT "INPUT SCALAR":INPUT R
20 PRINT "INPUT N,M":INPUT N,M
40 PRINT "INPUT MATRIX A"
50 FOR I=1 TO N
60 INPUT A(I,1),A(I,2),A(I,3),A(I,4),A(I,5),A(I,6),A(I,7),A(I,8)
      A(I,9),A(I,10)
80 NEXT I
90 IF Z=1 THEN 170: R=1
100 PRINT "INPUT MATRIX B"
110 FOR I=1 TO N
115 INPUT X(1),X(2),X(3),X(4),X(5),X(6),X(7),X(8),X(9),X(10)
120 FOR K=1 TO M
130 IF Z=2 THEN 140: X(K)=-X(K)
140 A(I,K)=A(I,K)+X(K)
150 NEXT K
160 NEXT I
170 FOR I=1 TO N:PRINT :FOR J=1 TO M
180 PRINT R*A(I,J):NEXT J:PRINT :NEXT I
190 END
```

```
1 DEFFN'0
5 DIM A(8,9),B(8,9),C(8)
10 PRINT HEX(03)
20 PRINT "INPUT N,M,P": INPUT N,P,M
40 PRINT "INPUT MATRIX A"
50 FOR I=1 TO N
60 INPUT A(I,1),A(I,2),A(I,3),A(I,4),A(I,5),A(I,6),A(I,7),A(I,8)
      A(I,9)
80 NEXT I: PRINT "INPUT MATRIX B"
90 FOR I=1 TO P
100 INPUT B(I,1),B(I,2),B(I,3),B(I,4),B(I,5),B(I,6),B(I,7),B(I,8)
      B(I,9)
120 NEXT I: FOR I=1 TO N: FOR J=1 TO M: FOR K=1 TO P
130 S=S+A(I,K)*B(K,J): NEXT K: C(J)=S: S=0: NEXT J
140 FOR J=1 TO M: A(I,J)=C(J): NEXT J: NEXT I
150 FOR I=1 TO N: PRINT : FOR J=1 TO M
160 PRINT A(I,J):NEXT J:PRINT :NEXT I
170 END
```

EQUATIONS

BLOCK NO.: 18

```
5 DIM A(9,9),X(8)
10 PRINT "INPUT NO. OF UNKNOWNNS, MAX. NO. OF ITERATIONS, DELTA"
15 INPUT N,M,D
20 PRINT "INPUT COEFF. MATRIX A AND THEN COLUMN MATRIX B"
100 FOR I=1 TO N+1
120 INPUT A(I,1),A(I,2),A(I,3),A(I,4),A(I,5),A(I,6),A(I,7),
      A(I,8),A(I,9)
140 NEXT I
170 FOR I=1 TO N
210 X(I)=A(N+1, I)
220 NEXT I
230 FOR K=1 TO M
310 T=0
320 FOR I=1 TO N
330 S=0
340 FOR J=1 TO N
350 S=S+A(I, J)*X(J)
360 NEXT J
370 B=X(I)
380 X(I)=(-S+A(I, I)*X(I)+A(N+1, I))/A(I, I)
390 IF ABS(X(I))<=ABS(B) THEN 420
400 IF ABS(B/X(I))<D THEN 440
410 GOTO 440
420 IF ABS(X(I)/B)<D THEN 440
430 T=T+1
440 NEXT I
450 IF T=N THEN 480
460 NEXT K
470 PRINT "CONVERGENCE TOO SLOW. LAST VALUES COMPUTED ARE:"
480 PRINT
485 FOR I=1 TO N
490 PRINT "X("; I; ") ="; X(I)
500 NEXT I
999 END
```



```

1 DIM A(10,14), X(13)
5 PRINT "NO. OF VARIABLES?": INPUT M
10 PRINT "NO. OF CONSTRAINTS?": INPUT N
15 PRINT "ENTER MATRIX A"
20 FOR I=2 TO N+1
25 INPUT A(I,1), A(I,2), A(I,3), A(I,4), A(I,5), A(I,6), A(I,7), A(I,8)
   A(I,9)
30 A(I, N+M+1)=A(I, M+2): A(I, M+2)=0: IF I=2 THEN 40
35 A(I, M+I-1)=A(I, M+1): A(I, M+1)=0
40 NEXT I
45 PRINT "ENTER OBJECTIVE FUNCTION"
50 INPUT A(1,1), A(1,2), A(1,3), A(1,4), A(1,5), A(1,6)
55 PRINT : R=1
60 FOR I=1 TO M: X(I)=1: NEXT I
65 FOR I=2 TO N+1
70 IF A(I, M+I-1) < -1 THEN 85: X(M+I-1)=1
75 FOR J=1 TO N+M: A(N+2, J)=A(N+2, J)-A(I, J): NEXT J
80 R=N+2
85 NEXT I
90 S, T=1
95 FOR I=2 TO N+M: IF A(R, I) < A(R, S) THEN 100: S=I
100 IF A(R, I) >= A(R, T) THEN 110: T=I
110 NEXT I
120 IF A(R, T) <= 0 THEN 140: IF R=1 THEN 300
130 IF A(R, S) > 1E-4 THEN 280: R=1: GOTO 90
140 S=1
150 FOR I=2 TO N+1: IF A(I, T) <= 0 THEN 190
160 Y=A(I, N+M+1)/A(I, T): IF S=1 THEN 180
170 IF Y >= A(S, N+M+1)/A(S, T) THEN 190
180 S=I
190 NEXT I
200 IF S=1 THEN 290
210 FOR I=1 TO N+M: IF X(I)=1 THEN 220: IF A(S, I)=1 THEN 230
220 NEXT I
230 X(I)=1: X(T)=0: Y=A(S, T)
240 FOR I=1 TO N+M+1: A(S, I)=A(S, I)/Y: NEXT I
250 FOR I=1 TO N+2: IF I=S THEN 270: Y=A(I, T)
260 FOR J=1 TO N+M+1: A(I, J)=A(I, J)-Y*A(S, J): NEXT J
270 NEXT I: GOTO 90
280 PRINT "INFEASIBLE": STOP
290 PRINT "UNBOUNDED": STOP
300 FOR J=1 TO M
310 IF X(J)=0 THEN 320: X(J)=0: GOTO 340
320 FOR I=2 TO N+1: IF A(I, J)=1 THEN 330: NEXT I
330 X(J)=A(I, N+M+1)
340 NEXT J
350 Y=A(1, N+M+1)
360 PRINT : PRINT "OBJ. FUNC. =": Y: PRINT
370 FOR I=1 TO M: PRINT "X(" ; I ; ")=" ; X(I): NEXT I
380 END

```

```
1 DIM A(5,5),B(5,5)
5 PRINT "ENTER N": INPUT N
10 PRINT : PRINT "ENTER MATRIX"
15 FOR I=1 TO N: PRINT : PRINT "COL": I
20 FOR J=1 TO N: INPUT A(J, I),B(J, I): NEXT J
25 NEXT I
30 D1, I1=1: D2=0
35 I3=I1: S=ABS(A(I1, I1))+ABS(B(I1, I1))
40 FOR I=I1 TO N: T=ABS(A(I, I1))+ABS(B(I, I1))
45 IF S>T THEN 50: I3=I: S=T
50 NEXT I: IF I3=I1 THEN 75
55 FOR J=1 TO N
60 S=-A(I1, J): A(I1, J)=A(I3, J): A(I3, J)=S
65 S1=-B(I1, J): B(I1, J)=B(I3, J): B(I3, J)=S1
70 NEXT J
75 I3=I1+1
80 FOR I=I3 TO N: S1=A(I1, I1)^2+B(I1, I1)^2
85 S=(A(I, I1)*A(I1, I1)+B(I, I1)*B(I1, I1))/S1
90 B(I, I1)=(A(I1, I1)*B(I, I1)-A(I, I1)*B(I1, I1))/S1
95 A(I, I1)=S: NEXT I
100 J2=I1-1: IF J2=0 THEN 150
110 FOR J=I3 TO N: FOR I=1 TO J2
120 A(I1, J)=A(I1, J)-A(I1, I)*A(I, J)+B(I1, I)*B(I, J)
130 B(I1, J)=B(I1, J)-B(I1, I)*A(I, J)-A(I1, I)*B(I, J)
140 NEXT I: NEXT J
150 J2=I1: I1=I1+1
160 FOR I=I1 TO N: FOR J=1 TO J2
170 A(I, I1)=A(I, I1)-A(I, J)*A(J, I1)+B(I, J)*B(J, I1)
180 B(I, I1)=B(I, I1)-B(I, J)*A(J, I1)-A(I, J)*B(J, I1)
190 NEXT J: NEXT I
200 IF I1<N THEN 35: I3=1: J2=INT(N/2)
210 IF N=2*J2 THEN 230: I3=0
220 D1=A(N, N): D2=B(N, N)
230 FOR I=1 TO J2: J=N-I+I3
240 S=A(I, I)*A(J, J)-B(I, I)*B(J, J)
250 S1=A(I, I)*B(J, J)+A(J, J)*B(I, I)
260 T=D1*S-D2*S1: D2=D2*S+D1*S1: D1=T
270 NEXT I: PRINT
280 PRINT "DETERMINANT: "; D1: "+I*": D2: PRINT
290 END
```

INVERSE HYPERBOLICS

BLOCK NO.: 21

```
5 SELECT R
10 PRINT "CODE, X? TO END PROGRAM INPUT 0, 0. "
20 INPUT N, X
25 IF N=0 THEN 240
30 IF N>3 THEN 140
40 S=(EXP(X)-EXP(-X))/2
50 C=(EXP(X)+EXP(-X))/2
60 IF N>1 THEN 90
70 PRINT "SINH(X)="; S
80 GOTO 210
90 IF N>2 THEN 120
100 PRINT "COSH(X)="; C
110 GOTO 210
120 PRINT "TANH(X)="; S/C
130 GOTO 210
140 IF N>4 THEN 170
150 PRINT "ARCSINH(X)="; LOG(X+SQR(X^2+1))
160 GOTO 210
170 IF N>5 THEN 200
180 PRINT "ARCCOSH(X)="; LOG(X+SQR(X^2-1))
190 GOTO 210
200 PRINT "ARCTANH(X)="; LOG((1+X)/(1-X))/2
210 PRINT
220 PRINT "CODE, X?"
230 GOTO 20
240 SELECT 0 :END
```

```
1 SELECT R
5 PRINT
10 PRINT "A, B? (TO END PROGRAM INPUT 0, 0)"
15 INPUT A, B
16 PRINT
20 IF A2+B2=0 THEN 999
25 DEFFNS(W)=(EXP(W)-EXP(-W))/2
30 DEFFNC(W)=(EXP(W)+EXP(-W))/2
35 X=SIN(A)*FNC(B)
40 Y=COS(A)*FNS(B)
45 PRINT "SIN(A+BI)="; X; "+"; Y; "I"
50 X=COS(A)*FNC(B)
55 Y=-SIN(A)*FNS(B)
60 PRINT "COS(A+BI)="; X; "+"; Y; "I"
65 D=COS(2*A)+FNC(2*B)
70 X=SIN(2*A)/D
75 Y=FNS(2*B)/D
80 PRINT "TAN(A+BI)="; X; "+"; Y; "I"
85 X=FNS(A)*COS(B)
90 Y=FNC(A)*SIN(B)
95 PRINT "SINH(A+BI)="; X; "+"; Y; "I"
100 X=FNC(A)*COS(B)
110 Y=FNS(A)*SIN(B)
120 PRINT "COSH(A+BI)="; X; "+"; Y; "I"
130 D=FNC(2*A)+COS(2*B)
140 X=FNS(2*A)/D
150 Y=SIN(2*B)/D
160 PRINT "TANH(A+BI)="; X; "+"; Y; "I"
170 PRINT
180 PRINT
190 PRINT "A, B?"
200 GOTO 15
999 SELECT D :END
```

```
1 SELECT R
5 PRINT
10 PRINT "ANGLE? (TO END PROGRAM INPUT 99999)"
15 INPUT X
20 IF X=99999 THEN 99
25 A=3600*180*X/3.1415927
30 D=INT(A/3600)
35 D1=INT(D/360)
40 M=INT((A-D*3600)/60)
45 PRINT D-360*D1, "DEG. "
50 PRINT M, "MIN. "
55 PRINT A-D*3600-M*60, "SEC. "
60 PRINT
65 PRINT "ANGLE?"
70 GOTO 15
99 SELECT D :END
```

```
5 SELECT R
10 PRINT
20 PRINT "ANGLE (DEG, MIN, SEC, CARRIAGE RETURN)?"
30 INPUT D, M, S
40 A=D+M/60+S/3600
50 R=INT(A/360)
60 PRINT "ANGLE="; A*. 17453293E-1-R*6. 2831853, "RADIANS"
70 PRINT
80 PRINT "MORE INPUT? (1=YES, 0=NO)"
90 INPUT D
100 IF D=1 THEN 10
110 SELECT D :END
```

```
10 SELECT R
11 READ N, A1, B1
12 PRINT "INPUT X (TO END PROGRAM INPUT 99999). "
13 INPUT X
14 IF X=99999 THEN 999
15 LET S=SIN(X)
20 LET C=COS(X)
25 LET F=A1*S+B1*C
30 LET U=S
35 LET V=C
40 FOR I=2 TO N
45     LET S=S*V+C*U
50     LET C=C*V-S*U
55     READ A, B
60     LET F=F+A*S+B*C
65 NEXT I
70 PRINT "F( ) X: ")=": F
75 RESTORE
80 PRINT
85 PRINT "INPUT X"
90 GOTO 13
100DATA 3, 1, 1, -2, 2, 3, 4
999 END
```

```
5 SELECT R
10 DIM A(3), S(3)
39 PRINT "PROBLEM TYPES: 0=END PROGRAM, 1=AAA, 2=ASA, 3=SSA, 4=SAS, 5
=SSS"
40 PRINT "INPUT PROBLEM TYPE"
50 INPUT Z
60 IF Z=0 THEN 999
70 IF Z=1 THEN 420
80 IF Z=2 THEN 390
90 IF Z=3 THEN 470
100 IF Z=4 THEN 350
300 PRINT "INPUT 'SIDE, SIDE, SIDE'"
305 INPUT S(1), S(2), S(3)
310 A(1)=(S(2)2+S(3)2-S(1)2)/2/S(2)/S(3)
320 A(1)=ATN(SQR(1-A(1)2)/A(1))
330 GOSUB 800
340 GOTO 40
350 PRINT "INPUT 'SIDE, ANGLE, SIDE'"
355 INPUT S(3), A(1), S(2)
360 S(1)=SQR(S(3)2+S(2)2-2*S(3)*S(2)*COS(A(1)))
370 GOSUB 800
380 GOTO 40
390 PRINT "INPUT 'ANGLE, SIDE, ANGLE'"
395 INPUT A(1), S(3), A(2)
400 A(3)=3.1415927-A(1)-A(2)
410 GOTO 440
420 PRINT "INPUT 'ANGLE, ANGLE, SIDE'"
425 INPUT A(3), A(2), S(3)
430 A(1)=3.1415927-A(2)-A(3)
440 S(1)=S(3)*SIN(A(1))/SIN(A(3))
450 S(2)=S(3)*SIN(A(2))/SIN(A(3))
460 GOSUB 825
465 GOTO 40
470 PRINT "INPUT 'SIDE, SIDE, ANGLE'"
475 INPUT S(1), S(2), A(1)
480 X=S(2)*SIN(A(1))
490 IF S(1)<X THEN 900
500 S(3)=SQR(S(2)2-X2)
502 IF S(1)>X THEN 510
504 GOSUB 800
506 GOTO 40
510 Z=SQR(S(1)2-X2)
530 S(3)=S(3)+Z
540 GOSUB 800
550 PRINT "ALTERNATE SOLUTION"
```



```
560 S(3)=S(3)-2*Z
565 A(3)=A(2)-A(1)
570 A(2)=3.1415927-A(2)
575 GOSUB 825
580 GOTO 40
800 A(2)=SIN(A(1))/S(1)*S(2)
810 A(2)=ATN(A(2)/SQRT(1-A(2)^2))
820 A(3)=3.1415927-A(1)-A(2)
825 PRINT
830 FOR I=1 TO 3
835 IF A(I)<0 THEN 901
840 PRINT "SIDE"; I; "="; S(I)
841 PRINT "OPPOSITE ANGLE="; A(I); "RADIANS"
850 NEXT I
860 PRINT
870 RETURN
900 PRINT
901 PRINT "NO SOLUTION"
902 PRINT
910 GOTO 40
999 SELECT D :END
```

```
5 SELECT R
11 READ N
12 PRINT "X", "Y", "R", "A"
15 IF N=-1 THEN 135
20 READ X, Y
25 IF X=0 THEN 45
30 IF Y=0 THEN 85
35 PRINT X, Y, SGN(X)*SQR(X2+Y2), ATN(Y/X)*180/3.1415927
40 GOTO 20
45 IF Y=0 THEN 75
47 PRINT X, Y, ABS(Y),
50 IF Y<0 THEN 65
55 PRINT 90
60 GOTO 20
65 PRINT 270
70 GOTO 20
75 PRINT X, Y, 0, 0
80 GOTO 20
85 PRINT X, Y, ABS(X),
90 IF X<1 THEN 120
100 PRINT 0
110 GOTO 20
120 PRINT 180
130 GOTO 20
135 READ R, A
136 LET Z=(A-INT(A/360)*360)*3.1415927/180
140 PRINT R*COS(Z), R*SIN(Z), R, A
150 GOTO 135
999 END
```

```
11 S=0
12 PRINT "N, X, Y"
15 INPUT N, X0, Y0
16 X=X0:Y=Y0
20 FOR I=1 TO N-1
22 PRINT "X"; I; ", Y"; I
25 INPUT X1, Y1
30 S=S+(X+X1)*(Y-Y1)
35 X=X1
40 Y=Y1
45 NEXT I
50 PRINT "A="; (S+(X+X0)*(Y+Y0))/2
55 PRINT "MORE INPUT (1=YES, 0=NO)"
60 INPUT I
65 IF I=1 THEN 11
70 END
```

```
5 PRINT "INPUT KNOWN POINTS (X1, Y1, X2, Y2, CARRIAGE RETURN)"
10 INPUT X1, Y1, X2, Y2
15 PRINT
20 PRINT "INPUT X-COORD. OF PT. TO BE INTERPOLATED. TO END "
25 PRINT "PROGRAM INPUT 99999. "
30 INPUT X
35 IF X=99999 THEN 99
40 PRINT "Y="; Y1+(Y2-Y1)/(X2-X1)*(X-X1)
45 PRINT
50 PRINT "X";
55 GOTO 30
99 END
```

```
3 DIM X(70),Y(70)
5 PRINT "INPUT NUMBER OF KNOWN POINTS, N"
10 INPUT N
15 PRINT "INPUT KNOWN POINTS (1 POINT TO A LINE) AS FOLLOWS: (X,
Y)"
20 FOR I=1 TO N
25 INPUT X(I),Y(I)
30 NEXT I
35 PRINT "INPUT X-COORDINATE OF POINT TO BE INTERPOLATED. TO END
"
36 PRINT "KEY 99999"
47 PRINT
60 PRINT "X"
65 INPUT Z
67 IF Z=99999 THEN 160
70 T=0
75 FOR K=1 TO N
80 S=1
85 FOR I=1 TO N
90 IF I=K THEN 100
95 S=S*(Z-X(I))/(X(K)-X(I))
100 NEXT I
110 T=T+S*Y(K)
120 NEXT K
130 PRINT "Y=";T
140 PRINT
150 GOTO 60
160 END
```

```
5 PRINT "***GREATEST COMMON DIVISOR OF TWO INTEGERS***"
10 PRINT
100 PRINT "INPUT (INTEGER, INTEGER). TO END PROGRAM INPUT (0, 0)"

110 INPUT A, B
115 IF A=0 THEN 999
120 PRINT
130 C=ABS(A)
140 D=ABS(B)
150 R=C-D*INT(C/D)
160 IF R=0 THEN 200
170 C=D
180 D=R
190 GOTO 150
200 PRINT "G. C. D. =" ; D
210 PRINT
220 PRINT "INPUT (INTEGER, INTEGER)"
230 GOTO 110
999END
```

INTEGER

BLOCK NO.: 32

```
100 PRINT "INPUT NUMBER TO BE FACTORED. TO END PROGRAM INPUT 0"

110 INPUT N
115 IF N=0 THEN 260
120 PRINT
130 PRINT "FACTORS"
135 PRINT SGN(N)
136 N=ABS(N)
140 FOR I=2 TO N
150     S=0
160     IF N/I<>INT(N/I) THEN 200
170     N=N/I
180     S=S+1
190 GOTO 160
200     IF S=0 THEN 220
210 PRINT I; " "; S
220 NEXT I
230 PRINT
240 PRINT "NUMBER";
250 GOTO 110
260 END
```

```
5 PRINT "INPUT (N,R). TO END PROGRAM INPUT VALUES<=0. "  
7 GOTO 20  
10 PRINT "INPUT (N,R)"  
20 INPUT N,R  
30 IF N<=0 THEN 999  
40 IF R<=0 THEN 999  
50 IF R<=N THEN 90  
60 PRINT "R MUST BE <=N"  
70 PRINT  
80 GOTO 10  
90 P=1  
100 FOR L=N-R+1 TO N  
110 IF 9.9E62/L>=P THEN 150  
120 PRINT "NO. OF PERMUTATIONS CANNOT BE COMPUTED DUE TO FLOATIN  
G"  
130 PRINT "POINT OVERFLOW"  
135 PRINT  
140 GOTO 10  
150 P=P*L  
160 NEXT L  
170 F=1  
175 IF R=1 THEN 210  
180 FOR I=2 TO R  
190 F=F*I  
200 NEXT I  
210 PRINT "NO. OF PERMUTATIONS IS ";P  
220 PRINT "NO. OF COMBINATIONS IS ";P/F  
230 PRINT  
240 GOTO 10  
999 END
```



```
10 PRINT "INPUT A, B. TO END PROGRAM INPUT 0, 0. "  
20 INPUT A, B  
25 IF A=0 THEN 70  
30 PRINT "LOG"; B; "TO BASE"; A; "="; LOG(B)/LOG(A)  
40 PRINT  
50 PRINT "INPUT A, B"  
60 GOTO 20  
70 END
```

```
5 PRINT "INPUT P, Q, R, S, T"
10 INPUT P, Q, R, S, T
15 PRINT "A="; P2-R
20 PRINT "B="; -2*P
25 PRINT "C="; 1
30 PRINT "D="; 2*P*Q-S
35 PRINT "E="; -2*Q
40 PRINT "F="; Q2-T
45 PRINT
50 PRINT "MORE INPUT? (1=YES, 0=NO)"
55 INPUT P
60 PRINT
65 IF P=1 THEN 5
70 END
```

EQUATION

```
5 PRINT
10 PRINT "ENTER P, Q, R, S, T"
15 INPUT P, Q, R, S, T
20 A=P2-R
25 D=2*P*Q-S
30 PRINT "P1="; P/A
35 PRINT "Q1="; -D/2/A
40 PRINT "R1="; (P/A)2-1/A
45 PRINT "S1="; -D*P/(A2)+2*Q/A
50 PRINT "T1="; (D/2/A)2-(Q2-T)/A
55 PRINT
60 PRINT "MORE INPUT? (1=YES, 0=NO)"
65 INPUT P
70 IF P=1 THEN 5
75 END
```

```
5 PRINT "ENTER A, B, C, D, E, F"
10 INPUT A, B, C, D, E, F
15 PRINT "P=";  $-B/A/2$ 
20 PRINT "Q=";  $-D/A/2$ 
25 PRINT "R=";  $(B/A/2)^2 - C/A$ 
30 PRINT "S=";  $(B*D/2 - E)/A$ 
35 PRINT "T=";  $(D/A/2)^2 - F/A$ 
40 PRINT
45 PRINT "MORE INPUT? (1=YES, 0=NO)"
50 INPUT A
55 PRINT
60 IF A=1 THEN 5
65 END
```

NUMBERS

BLOCK NO.: 39

```
11 READ N, X, Y
15 FOR I=1 TO N-1
20 READ M, V, W
22 A=X
25 IF M=2 THEN 200
30 IF M=3 THEN 300
35 IF M=4 THEN 400
100 X=X+V
110 Y=Y+W
120 GOTO 500
200 X=X-V
210 Y=Y-W
220 GOTO 500
300 X=X+V-Y+W
310 Y=A+W+V+Y
320 GOTO 500
400 X=(X*V+Y*W)/(V^2+W^2)
410 Y=(V*Y-A*W)/(V^2+W^2)
500 NEXT I
510 PRINT X, Y; "I"
999END
```

```
5 PRINT "A, B, Z, N? (TO END PROGRAM INPUT 0, 0, 0, 0)"
10 INPUT A, B, Z, N
15 IF A^2+B^2=0 THEN 99
20 S=1
25 FOR I=N TO 1 STEP -1
30 S=1+(A+I-1)*Z*S/((B+I-1)*I)
35 NEXT I
40 PRINT "M="; S
45 PRINT
50 PRINT "A, B, Z, N?"
55 GOTO 10
99 END
```

```
5 PRINT
7 PRINT "INPUT COMPLEX NO. (A, B, CARRIAGE RETURN)"
10 INPUT A, B
15 IF B<0 THEN 50
20 IF B>0 THEN 50
25 IF A<0 THEN 40
30 PRINT "ANS. : "; SQR(A)
35 GOTO 5
40 PRINT "ANS. : "; SQR(-A); "I"
45 GOTO 5
50 R=SQR(A2+B2)
55 PRINT "ANS. : "; SQR((A+R)/2); SGN(B)*SQR((R-A)/2); "I"
60 GOTO 5
65 END
```

```
10 PRINT "INPUT I,X. TO END PROGRAM INPUT 0,0"
20 INPUT I,X
25 IF I=0 THEN 190
30 X1=(X/2)^2
40 S,J,T,L=1
50 K=-1
70 T=T*X1/J/(I+J)
80 S=S+K*T
90 IF T<1E-6 THEN 130
100 K=-K
110 J=J+1
120 GOTO 70
130 FOR K=1 TO I
140 L=L*K
150 NEXT K
160 PRINT "ANS. :"(X/2)^I/L*S
170 PRINT
180 PRINT "INPUT I,X"
185 GOTO 20
190 END
```



```
10 PRINT "INPUT X. TO END PROGRAM INPUT 0. "  
20 INPUT X  
30 IF X=0 THEN 999  
40 IF INT(X)/ABS(X)<0-1 THEN 70  
50 PRINT "GAMMA(X) UNDEFINED"  
60 GOTO 910  
70 IF X<2 THEN 200  
100 P=INT(X)-1  
110 X=X-P  
120 GOSUB 500  
130 GOSUB 600  
140 X=S*T  
150 GOTO 900  
200 IF X<1 THEN 300  
210 GOSUB 600  
220 X=T  
230 GOTO 900  
300 IF X<0 THEN 400  
310 Y=X  
320 X=X+1  
330 GOSUB 600  
340 X=T/Y  
350 GOTO 900  
400 P=ABS(INT(X))+1  
410 GOSUB 500  
420 X=X+P  
430 Y=X  
440 X=X+1  
450 GOSUB 600  
460 X=T/Y/S  
470 GOTO 900  
500 S=1  
510 FOR I=0 TO P-1  
520 S=S*(X+I)  
530 NEXT I  
540 RETURN  
600 T, W=1  
610 X=X-1  
620 FOR I=1 TO 8  
630 W=W*X  
640 READ B  
650 T=T+W*B  
660 NEXT I  
670 RETURN  
680 DATA -. 57719165, . 98820589, -. 89705694  
690 DATA . 91820686, -. 75670408, . 48219939  
700 DATA -. 19352782, . 03586834  
900 PRINT "GAMMA(X)="; X  
910 PRINT  
920 PRINT "X";  
930 RESTORE  
940 GOTO 20  
999 END
```

TITLE: FOURIER ANALYSIS

PROGRAM NO.: PS.02-2200.01A-00FI-43-0

TAPE NO.: 701-0119

(DEFINED FUNCTION)

BLOCK NO.: 43

```
5 N=35
10 DIM Y(71),A(35),B(35)
15 FOR I=1 TO 2*N+1: Y(I)=FNC((I-1)*.88495568E-1): NEXT I
20 PRINT "INPUT MAXIMUM HARMONIC TO BE CALCULATED (<N>):"; INPUT M
25 PRINT : PRINT "HARMONIC", "COSINE COEFF.", "SINE COEFF."
30 C3=2/(2*N+1): K1=3.1415927*C3
35 S1=SIN(K1): C1=COS(K1)
40 J,C=1: S=0: W=Y(1)
45 U1,U2=0: I=2*N+1
50 U3=Y(I)+2*C*U1-U2: U2=U1: U1=U3
55 I=I-1: IF I>1 THEN 50
60 A(J)=C3*(W+C*U1-U2): IF J<>1 THEN 70
65 A(1)=A(1)*.5
70 B(J)=C3*S*U1: PRINT J-1,A(J),B(J)
75 IF J=M+1 THEN 90
80 Q=C1*C-S1*S: S=C1*S+S1*C: C=Q
85 J=J+1: GOTO 45
90 PRINT
95 PRINT "INPUT X-VALUE TO BE EVALUATED. TO END PROGRAM INPUT 99
999"
100 INPUT W: IF W=99999 THEN 999
110 I=0
120 FOR J=1 TO M+1: I=I+A(J)*COS((J-1)*W)+B(J)*SIN((J-1)*W): NEX
T J
130 PRINT "Y="; I: PRINT :PRINT :PRINT "X": GOTO 100
999 END
```

(TABULATED FUNCTION)

BLOCK NO.: 44

```
10 DIM Y(32),A(15),B(15)
15 PRINT "INPUT N": INPUT N
20 PRINT "INPUT VALUES OF FUNCTION (4/LINE)"
25 FOR I=0 TO INT((N+2)/2)-1
30 INPUT Y(4*I+1),Y(4*I+2),Y(4*I+3),Y(4*I+4): NEXT I
35 PRINT "INPUT MAXIMUM HARMONIC TO BE CALCULATED (<=N)": INPUT
M
40 PRINT : PRINT "HARMONIC      COSINE COEFF.   SINE COEFF. "
45 C3=2/(2*N+1): K1=3.1415927*C3
50 S1=SIN(K1): C1=COS(K1)
55 J,C=1: S=0: W=Y(1)
60 U1,U2=0: I=2*N+1
65 U3=Y(I)+2*C*U1-U2: U2=U1: U1=U3
70 I=I-1: IF I>1 THEN 65
75 A(J)=C3*(W+C*U1-U2): IF J<>1 THEN 85
80 A(1)=A(1)*.5
85 B(J)=C3*S*U1: PRINT J-1,A(J),B(J)
87 IF J>=M+1 THEN 100
90 Q=C1*C-S1*S: S=C1*S+S1*C: C=Q
95 J=J+1: GOTO 60
100 PRINT
110 PRINT "INPUT X-VALUE TO BE EVALUATED. TO END PROGRAM INPUT 9
9999"
120 INPUT W: IF W=99999 THEN 999
130 I=0
140 FOR J=1 TO M+1: I=I+A(J)*COS((J-1)*W)+B(J)*SIN((J-1)*W):NEXT
J
150 PRINT "Y=";I: PRINT : PRINT "INPUT X": GOTO 120
999 END
```



Jack Jarvis & Company, Inc.
707 S. W. Washington Street
Portland, Oregon 97205
Telephone: (503) 224-7838

**WANG LABORATORIES
(CANADA) LTD.**

180 Duncan Mill Road
Don Mills, Ontario M3B 1Z6
TELEPHONE (416) 449-7890
TELEX 06-21-7549

WANG EUROPE, S.A.

Buurtweg 13
9412 Ottergem
Belgium
TELEPHONE: 053/74514
TELEX: 26077

WANG ELECTRONICS LTD.

40-44 High Street
Northwood, Middlesex, England
TELEPHONE Northwood 27677

WANG FRANCE SARL

47, Rue de la Chapelle
Paris 18, France
TELEPHONE 203.27.94 or 203.25.94

WANG LABORATORIES GMBH

Moselstrasse No. 4
6000 Frankfurt am Main
West Germany
TELEPHONE (611) 23-00-40

WANG SKANDINAVISKA AB

Fredsgaten 17
S-172-23
Sundbyberg 1, Sweden
TELEPHONE 08-98-12-45

WANG NEDERLAND B.V.

Damstraat 2
Utrecht, Netherlands
TELEPHONE 030-930947

WANG PACIFIC LTD.

61, King Yip Street, 1st Floor
Kwun Tong, Kowloon, Hong Kong
TELEPHONE 3-434231/2

WANG INDUSTRIAL CO., LTD.

110-118 Kuang-Fu N. Rd.
Taipei, Taiwan
Republic of China
TELEPHONE 784181-3

WANG GESELLSCHAFT MBH

Grinzing Allee 16
1190 Vienna 19
Austria
TELEPHONE (0222) 32.42.43

WANG COMPUTER PTY. LTD.

25 Bridge Street
Pymble, NSW 2073
Australia
TELEPHONE 449-6388

**WANG INTERNATIONAL
TRADE, INC.**

836 North Street
Tewksbury, Massachusetts 01876
TELEPHONE (617) 851-4111
TWX 710-343-6769
TELEX 94-7421

PHI COMPUTER SERVICES

836 North Street
Tewksbury, Massachusetts 01876
TELEPHONE (617) 851-4111
TWX 710-343-6769
TELEX 94-7421
24 Mill Street
Arlington, Massachusetts 02174
TELEPHONE (617) 648-8550

WANG

LABORATORIES, INC.

836 NORTH STREET, TEWKSBURY, MASSACHUSETTS 01876, TEL (617) 851-4111, TWX 710 343-6769, TELEX 94-7421

Printed in U.S.A.
700-3132
7-73-1.5M
Book Price \$25.00
Package Price \$250.00